The particle in a box and the uncertainty principle

Marc R. Roussel

Department of Chemistry and Biochemistry

University of Lethbridge

We start by defining the wavefunction. It depends on three "variables", namely the quantum number n, the length of the box L and the position x.

```
> psi := (n,L,x) -> sqrt(2/L)*sin(n*Pi*x/L);
```

$$\psi := (n, L, x) \to \sqrt{\frac{2}{L}}\,\sin\left(\frac{n\,\pi\,x}{L}\right)$$

The uncertainty principle says that $\Delta x \Delta p \geq$ (hbar)/2. Δx and Δp are calculated from $<x>$, $<x^2>$, etc. I'll start by calculating the average x, which is found by evaluating the integral

```
> Int(psi(1,L,x)*x*psi(1,L,x),x=0..L);
```

$$\int_0^L \frac{2\sin\left(\frac{\pi\,x}{L}\right)^2 x}{L}\,dx$$

```
> avg_x := value(%);
```

$$avg\_x := \frac{L}{2}$$

Note that I carried out the integration in two steps using the inert "Int" so that you could see the integral being evaluated before we actually obtained a value. Normally, you would just use "int" and get the answer right away. Also, I inserted the value n=1 into the wavefunction through the argument of the Maple function I created.

The answer we get makes perfect sense: The average x is in the middle of the box.

Now let's calculate $<x^2>$:

```
> Int(psi(1,L,x)*x^2*psi(1,L,x),x=0..L);
```

$$\int_0^L \frac{2\sin\left(\frac{\pi\,x}{L}\right)^2 x^2}{L}\,dx$$

```
> avg_x2 := value(%);
```

$$avg\_x2 := \frac{(-3+2\,\pi^2)\,L^2}{6\,\pi^2}$$

The average of p and of p^2 are a little more complicated to evaluate since p = -i(hbar)d/dx:

```
> Int(psi(1,L,x)*(-I)*hbar*diff(psi(1,L,x),x),x=0..L);
```

$$\int_0^L \frac{-2\,I \sin\left(\frac{\pi\,x}{L}\right) hbar \cos\left(\frac{\pi\,x}{L}\right)\pi}{L^2}\,dx$$

Note carefully how I entered this expression in Maple.

```
> avg_p := value(%);
```

$$avg\_p := 0$$

...another common-sense value.

The operator p^2 = -(hbar)^2 d^2/dx^2, so <p^2> is

```
> Int(psi(1,L,x)*(-hbar^2)*diff(psi(1,L,x),x$2),x=0..L);
```

$$\int_0^L \frac{2\,\sin\left(\frac{\pi\,x}{L}\right)^2 hbar^2\,\pi^2}{L^3}\,dx$$

```
> avg_p2 := value(%);
```

$$avg\_p2 := \frac{\pi^2\,hbar^2}{L^2}$$

Now to calculate the standard deviations:

```
> Delta_x := sqrt(avg_x2-avg_x^2);
```

$$Delta\_x := \frac{\sqrt{\frac{6\,(-3+2\,\pi^2)\,L^2}{\pi^2} - 9\,L^2}}{6}$$

This will simplify if we inform Maple that L is positive.

```
> assume(L>0);
> Delta_x := simplify(Delta_x);
```

$$Delta\_x := \frac{\sqrt{3}\,L\sim\sqrt{-6+\pi^2}}{6\,\pi}$$

```
> Delta_p := sqrt(avg_p2-avg_p^2);
```

$$Delta\_p := \frac{\pi\sqrt{hbar^2}}{L\sim}$$

Again, Maple needs to know that hbar is positive to simplify this expression:

```
> assume(hbar>0);
> Delta_p := simplify(Delta_p);
```

$$Delta\_p := \frac{\pi\, hbar\sim}{L\sim}$$

The uncertainty product is

```
> Delta_x*Delta_p;
```

$$\frac{\sqrt{3}\,\sqrt{-6+\pi^2}\; hbar\sim}{6}$$

```
> evalf(%);
```

$$0.5678618088\; hbar\sim$$

which is bigger than (hbar)/2, as Heisenberg told us it should be.